

```

*****;
***                               ***;
*** Random.sas - Randomizes subjects and assigns therapist ***;
***                               ***;
*** Created by: DLF                ***;
*****;

```

```

*%let root=H:\Presentations\DslugApr08;
%let root=C:\Projects\UCHSC\DslugApr08;
option nfmterr;
*options mlogic mprint macrogen; * For Debugging *;

```

```

proc format;
  value TxGroup 1='CBT' 2='SNDT';
  value Site     1='Pitt' 2='Boston';
  value logical 0='No' 1='Yes';
  value Strata1x 1='Crohns' 2='Colitus';
  value Strata2x 1='Sub-syndromal' 2='Syndromal';
  value Strata3x 1='9-12 Years' 2='13-17 Years';
  value Strata   1='Crohns/Sub/9-12'
                2='Crohns/Sub/13-17'
                3='Crohns/Syindr/9-12'
                4='Crohns/Syindr/13-17'
                5='Colitus/Sub/9-12'
                6='Colitus/Sub/13-17'
                7='Colitus/Syindr/9-12'
                8='Colitus/Syindr/13-17'
                ;

```

```

run;

%macro Login; *** User login ***;
  *** Declares global macro variables USER, PW and XXX ***;
  %global USER PW Site;
  *** Queries User Name, PW and Site ***;
  %window interface1 color=yellow
  #5 @15 "**** Randomization Program ****"
  #7 @15 "Please type in your username and password"
  #9 @15 'User Name:      ' +1 USER 6 ATTR=UNDERLINE REQUIRED=YES
  #11 @15 'Password:      ' +1 PW 6 ATTR=UNDERLINE REQUIRED=YES DISPLAY=NO
  #13 @15 'Site (1=Pitt,2=Boston)' +1 SITE 1 ATTR=UNDERLINE REQUIRED=YES
  #24 @15 "Press ENTER to continue";
  %display interface1;
%mend;

```

```

%macro GetData;
  *** Reads Previous Randomization Assignments ***;
  PROC IMPORT OUT= WORK.RandomAssign
    DATATABLE= "RandomAssign"
    DBMS=ACCESS REPLACE;
  DATABASE=&root\RECRUITMENT.mdb;
  USEDATE=Yes;
RUN;
proc sort data=work.RandomAssign;
  by RandID;
run;
*** Number of subjects Randomized per Site ***;
Proc summary data=work.randomassign nway;

```

```

class Site;
var RandID;
output out=work.SiteN max=NSite;
run;
*proc print data=work.SiteN; run;
%mend;

```

```

%macro NewData;
  *** Declares global macro variables ***;
  %global StudyID Initials Strata1 Strata2 AgeYr Confirm;
  *** Queries Data on Child to be Randomized ***;
  %window interface2 color=grey
  #5 @15 "**** Randomization Program (Cont) ****"
  #7 @15 "Please type in the following information"
  #8 @15 'Childs Study ID:      ' +1 StudyID 3 ATTR=UNDERLINE REQUIRED=YES
  #9 @15 'Childs Initials:      ' +1 Initials 3 ATTR=UNDERLINE REQUIRED=YES
  #10 @15 'Diagnosis (1=Crohns D 2=Colitus):' +1 Strata1 1 ATTR=UNDERLINE REQUIRED=YES
  #11 @15 'Sub-Syndromal(1) or Syndormal(2)' +1 Strata2 1 ATTR=UNDERLINE REQUIRED=YES
  #12 @15 'Age in Years (9-17 only):      ' +1 AgeYr 2 ATTR=UNDERLINE REQUIRED=YES
  #15 @15 'Have all eligibility requirement been satisfied?'
        +2 Confirm 1 ATTR=UNDERLINE REQUIRED=Yes
  #20 @15 "Press ENTER to continue";
  %display interface2;
  *** Creates Record for New Child ***;
  data work.dummy; * Initializes variables when not assigned*;
  site=1; Nsite=1000; output;
  site=2; Nsite=2000; output;
run;
data work.new; * Merges number previously randomized at site *;
  update work.dummy work.SiteN;
  by Site;
  if Site=&site; * Selects Total for Site *;

  *** Computes ID and Assigns variables ***;
  RandID=(Nsite+1);
  StudyID=&StudyID+0;
  Initials="&Initials";

  *** Creates messages when entries are incorrect ***;
  window ReEnter color=grey
  group=problem1
  #3 @15 'Diagnosis must be either 1=Crohns D 2=Colitus'
  #4 @15 "Delete and ReEnter:" +1 Strata1 1. ATTR=UNDERLINE REQUIRED=YES
  #6 @15 "Press ENTER to continue"
  group=problem2
  #9 @15 'Severity must be either 1=Sub-Syndromal or 2=Syndormal'
  #10 @15 "Delete and ReEnter:" +1 Strata2 1. ATTR=UNDERLINE REQUIRED=YES
  #12 @15 "Press ENTER to continue"
  group=problem3
  #15 @15 'Age in Years must be between 9 and 17'
  #16 @15 'Press Delete x2 and ReEnter:' +1 AgeYr 2. ATTR=UNDERLINE REQUIRED=YES
  #18 @15 "Press ENTER to continue" ;

  *** Compute Strata and Send Message if a problem ***;
  Strata1=&Strata1+0; * Site *;
  if &Strata1 ^in(1,2) then display ReEnter.Problem1;

```

```
Strata2=&Strata2+0; * Severity *;
if &Strata2 ^in(1,2) then display ReEnter.Problem2;
```

```
AgeYr=&AgeYr+0;
if &AgeYr <9 or &AgeYr>17 then display ReEnter.Problem3;
Strata3=(9<=AgeYr<=12)+(13<=AgeYr<=17)*2; * Age Group *;
```

```
*** Combines 3 factors to get overall Strata ***;
if Strata eq . then Strata=(Strata1-1)*4+(Strata2-1)*2+Strata3;
```

```
format Strata1 strata1x. strata2 strata2x. Strata3 strata3x.
      Strata strata.;
```

```
drop _type_ _Freq_;
*Proc print data=work.new;
run;
```

```
%mend;
```

```
%macro Balance;
```

```
*** Calculates current balance for each strata ***;
```

```
* Creates a dataset with N=0 for All Strata and Tx Groups *;
```

```
data work.Bdummy;
do site=1 to 2;
do strata=1 to 8;
do txgroup=1 to 2;
n=0; output;
end; end; end;
```

```
* Sums the actual numbers in all Strata and Tx Groups *;
```

```
Proc summary data=work.randomassign nway;
```

```
class site strata TxGroup;
var RandID;
output out=work.assign n=N;
```

```
run;
```

```
* Updates the dataset with N=0 *;
```

```
data work.assign;
update work.Bdummy work.assign;
```

```
by site strata TxGroup;
```

```
proc sort data=work.assign;
```

```
by site strata TxGroup;
```

```
run;
```

```
* Reshapes the dataset *;
```

```
proc transpose data=work.assign out=work.balance prefix=TxGroup;
```

```
by site strata;
```

```
id TxGroup;
```

```
Var N;
```

```
run;
```

```
*** Merges Balance with Data from current subject ***;
```

```
data work.random;
```

```
merge work.new(in=ok)
      work.balance(Keep=site strata TxGroup1 TxGroup2);
```

```
by site strata;
```

```
*** Selects patient to be randomized ***;
```

```
if ok;
```

```
*** Calculates balance within strata ***;
```

```
if TxGroup1 eq . then TxGroup1=0;
```

```
if TxGroup2 eq . then TxGroup2=0;
```

```
balance=TxGroup1-TxGroup2;
```

```
*** Generates random number and assignment ***;
```

```
random=ranuni(int(Time())); * Random Number for TxGroup *;
```

```
if random le (.5-balance*.2) then TxGroup=1;
```

```
else TxGroup=2;
```

```
format TxGroup TxGroup.;
```

```
*** Randomization Date and User for documentation ***;
```

```
RandDate=today();
```

```
RandUser="&user";
```

```
*** Creates Macro variable for Therapist Time ***;
```

```
if txgroup eq 1 then %let TherProp=TherCBT;;
```

```
if txgroup eq 2 then %let TherProp=TherSNDT;;
```

```
*proc print data=work.random;
```

```
run;
```

```
%mend;
```

```
%macro AssgnTher;
```

```
*** Reads Dataset of Therapists ***;
```

```
PROC IMPORT OUT= WORK.Therapist
```

```
DATATABLE= "Therapists"
```

```
DBMS=ACCESS REPLACE;
```

```
DATABASE="&root\RECRUITMENT.mdb";
```

```
RUN;
```

```
proc sort data=work.Therapist;
```

```
by TherInit;
```

```
run;
```

```
*** Sums Current Active Assignments ***;
```

```
proc summary data=work.randomassign nway;
```

```
class TherInit;
```

```
var active;
```

```
output out=work.temp sum=Active;
```

```
run;
```

```
data work.Ther2;
```

```
merge work.therapist work.temp(keep=TherInit Active);
```

```
by TherInit;
```

```
if Active eq . then Active=0;
```

```
RandUser="&user"; * Added so that merge will work *;
```

```
data work.ther2b;
```

```
merge work.ther2 work.random(keep=txgroup randuser);
```

```
by RAndUser;
```

```
if TxGroup eq 1 then Avail=max(Thermax*TherCBT-Active-.5,0);
```

```
if TxGroup eq 2 then Avail=max(Thermax*TherSNDT-Active-.5,0);
```

```
if TherID eq 1 then Avail=.001; * Small positive value for ES *;
```

```
rename Active=TherActive;
```

```
*proc print data=work.Ther2b;run;
```

```
proc sql; * Calculates Proportion of Total Availability *;
```

```
create table work.Ther3 as
```

```
select *,
```

```
(Avail/sum(Avail)) as PAvail label='Prop of Total Availability'
```

```
from work.Ther2b;
```

```
quit;
```

```
*proc print data=work.Ther3;run;
```

```
*** Compares Random Variable to Cumulative Proportion *;
```

```

data work.Ther4;
  set work.Ther3;
  * Calculates Cumulative Distribution of Available Time *;
  retain CAvail1 CAvail2;
  if _N_ eq 1 then CAvail1=0;
  else Cavail1=Cavail2;
  Cavail2=Cavail1+PAvail;
  * Generates Random Number and Selects Therapist *;
  retain random2;
  if _N_ eq 1 then random2=ranuni(int(time()));
  Select=(CAvail1<Random2<=CAvail2);
  if Select then output; * Outputs Selected Therapist *;
  label CAvail1='Lower Boundary'
        CAvail2='Upper Boundary';

*proc print data=work.ther4;run;

*** Selects address of assigned Therapist and Creates a Macro Variable ***;
data addcontacts;
  merge work.Therapist work.Ther4(in=ok);
  by TherInit;
  if ok;
  CALL SYMPUT('TherLIST',TherEmail);
run;
data addcontacts;
  merge addcontacts work.random;
  rename Select=Active;
  *** Displays the results ***;
window ShowRan color=yellow
  #5 @10 "The assigment for:"
      +1 Initials PROTECT=YES AUTOSKIP=YES
      +2 "Study ID:" +1 StudyID PROTECT=YES AUTOSKIP=YES
      +2 "is" +1 TxGroup PROTECT=YES AUTOSKIP=YES
  #7 @10 "Diagnosis:" +1 Strata1 PROTECT=YES AUTOSKIP=YES
      +3 "Severity:" +1 Strata2 PROTECT=YES AUTOSKIP=YES
      +3 "Age (yrs):" +1 AgeYr PROTECT=YES AUTOSKIP=YES
  #10 @10 "The assigned therapist is:"
      +2 TherFirst PROTECT=YES AUTOSKIP=YES
      +2 TherLast PROTECT=YES AUTOSKIP=YES
  #12 @15 "who currently has"
      +1 TherActive PROTECT=YES AUTOSKIP=YES
      +1 "patients"
  #18 @10 "Type ENTER to continue";
display ShowRan;
run;
*** Sends E-mail Message to Therapist or PV if Control ***;
FILENAME MAIL EMAIL EMAILSYS=MAPI
  TO=("&TherLIST")
  FROM=("diane.fairclough@uchsc.edu")
  SUBJECT="New Randomization"
  ;
Data _NULL_;
  Set Addcontacts;
  File MAIL;
  Put @1 TherFirst "-";
  Put @4 "You have been assigned patient"

```

```

      +1 Initials
      +1 "(Study ID " StudyID)";
  Put @4 "who has been assigned to the " TxGroup txgroup.;
  Put @14 "Diagnosis:" +1 Strata1;
  Put @14 "Severity:" +1 Strata2;
  Put @14 "Age (yrs):" +1 AgeYr;
run;
%mend;

%macro Update;
*** Writes randomization assignment to ACCESS tables ***;
data work.update;
  set work.randomassign
      work.addcontacts(keep=StudyID RandID Initials Site strata
                      Strata1 strata2 Ageyr TxGroup
                      RandDate RandUser TherInit Active);
  by RandID;
  format randdate mmdyy10.;
*proc print data=work.update;
run;
PROC EXPORT DATA= WORK.UPDATE
  OUTTABLE= "RandomAssign"
  DBMS=ACCESS REPLACE;
  DATABASE="&root\RECRUITMENT.mdb";
RUN;
%mend;

%macro RunAll;
%Login;
*** Check Password ***;
%if %upcase(&PW)=RANDOM %then %do;
  %GetData; * Reads data from ACCESS Database *;
  %Newdata; * Queries data on new subject *;
  *** Run Randomization for those who have not been randomized ***;
  %if %upcase(&Confirm)=Y %then %do;
    %Balance;
    %AssgnTher;
    %Update;
  %end;
%end;
proc tabulate data=work.update;
  title 'Summary of All Randomizations';
  class site strata txgroup TherInit Active;
  table site*strata all, txgroup*N*F=4./rts=40;
  table (TherInit) all, Active*(TxGroup='Tx' All)*N*F=4./rts=40;
  format txgroup txgroup. strata strata. site site. Active logical.;
run;
%mend;

%RunAll;

```